
Ejemplos de programas de Funciones y Procedimientos 
 
6. Programa juego para adivinar números: genera un número aleatorio entre 1 y 100 y le propone 

adivinarlo al usuario 
 
(************************************************************) 
(* Programa Adivina_el_numero                               *) 
(* Juego para adivinar un numero del 1 al 100               *) 
(* Division de Informatica Ind. ETSI Industriales. UPM      *) 
(************************************************************) 
program adivina_el_numero; 
uses crt; 
var  numero,i,j : integer; 
begin 
clrscr; 
randomize; 
numero:=random(99)+1; 
j:=0; 
repeat 
  j:=j+1; 
  write('Es tu ',j,'º oportunidad. ¿Cual es el número? '); 
  readln(i); 
  if numero>i then writeln('Es mayor.':48) 
              else if numero<>i 
                then writeln('Es menor.':48) 
until numero=i; 
writeln('¡¡¡ Has acertado !!!'); 
repeat until keypressed 
end. 
 
 
7. Programa que calcula las raíces de un polinomio dentro de un intervalo determinado 
 
(************************************************************) 
(* Programa raices                                          *) 
(* Calcula raices de un polinomio por el metodo de la       *) 
(* biseccion de un bucle for                                *) 
(* Division de Informatica Ind. ETSI Industriales. UPM      *) 
(************************************************************) 
program raices; 
var  coef    : array[0..20] of real; 
     l,m,n   : real; 
     grado,I : integer; 
function f(a:real):real; 
{ calcula el valor del polinomio en a por la regla de Horner } 
     var  g : real; 
      begin 
      g:=0; 
      for I:=grado downto 0 do g:=g*a+coef[I]; 
      f:=g 
      end; 
begin 
write('Introduce el grado del polinomio = '); 
readln(grado); 
for I:=0 to grado do 
 begin 
    write('Introduce el coeficiente ',I,' del polinomio : '); 
      readln(coef[I]) 
      end; 
repeat 
  writeln('Introduce intervalo busqueda de raices: Ej: -1 1'); 
  writeln('Uno debe dar un valor negativo y el otro positivo'); 
  readln(l,m); 
  write('f(',l:6:4,') = ',f(l):6:4); 
  writeln(' ; f(',m:6:4,') = ',f(m):6:4); 

until (f(l)*f(m)<0); 
while abs(m-l)>0.000001 do 
  begin 
  n:=(m+l)/2; 
  if f(n)*f(l)<=0 then m:=n 
      else l:=n 
  end; 
if abs(f(n))<0.001 
 then writeln('La raiz esta en x = ',m:10:6) 
      else writeln('No encuentro ninguna raiz.') 
end. 
 
8. Programa que calcula la integral definida de la función f(x)= 1/(1+x2) dentro de un intervalo real 

(a,b) empleando la regla de Simpson:  

 donde h = (b−a)/2m. Simplificando la expresión de la derecha: 

 
(************************************************************) 
(* Programa Simpson                                         *) 
(* Integracion definida de una funcion utilizando la regla  *) 
(* de Simpson                                               *) 
(* Division de Informatica Ind. ETSI Industriales. UPM      *) 
(************************************************************) 
program simpson; 
var  a,b:real; p:integer; 
function f(x:real):real;   { Declara la funcion a integrar} 
  begin 
  f:=1/(1+x*x) 
  end; 
function i(a,b:real;p:integer):real; { Declara la funcion i } 
     var h,s:real; j:integer; 
     begin 
     h:=(b-a)/p; 
     s:=f(a); 
     for j:=1 to p do s:=s+4*f(a+h*(j-1/2))+2*f(a+h*j); 
     s:=s-f(b); 
     i:=h*s/6; 
     end; 
begin                                  { Programa principal } 
write('a = '); 
readln(a);            { Inicio del intervalo de integracion } 
write('b = '); 
readln(b);               { Fin del intervalo de integracion } 
write('Nº de subintervalos, p = '); 
readln(p); 
writeln('i = ',i(a,b,p)); 
readln 
end. 

